NEBRASKA

Power Analysis and Examples in SAS

Ryan Lalicker

SC3L

- Location: Hardin Hall North Wing, 346A
- Email: stathelp@unl.edu
- Workshop GitHub: https://github.com/StatHelpUNL/Workshop_25Spring

SAS OnDemand

- https://www.sas.com/en_us/software/on-demand-for-academics.html
- Free for students

What is “power”?

- Let’s say we collect data for some experiment
- We have some null and alternative hypothesis - H_0 and H_a
- Using the data, we either
  1. Fail to reject H_0 or
  2. Reject H_0
- In reality, either
  1. H_0 is true or
  2. H_0 is false

Possible outcomes of our decision

		Truth	
		H_0 True	H_0 False
Decision	Fail to Reject H_0	Correct Decision	Type II Error (False Negative)
	Reject H_0	Type I Error (False Positive)	Correct Decision

Recall significance/confidence levels

- We decide to reject/fail to reject H_0 if the p-value found is less than the significance level
  - α
  - Common levels are 0.01, 0.05, 0.10
- The confidence level is the remaining percentage
  - $1 - \alpha$
- $P(\text{Reject } H_0 | H_0 \text{ True}) = P(\text{Type I Error}) = \alpha$
- $P(\text{Failing to Reject } H_0 | H_0 \text{ True}) = 1 - \alpha$

Possible outcomes of our decision

		Truth	
		H_0 True	H_0 False
Decision	Fail to Reject H_0	Correct Decision **Prob. = $1 - \alpha$**	Type II Error (False Negative) **Prob. = ?**
	Reject H_0	Type I Error (False Positive) **Prob. = α**	Correct Decision **Prob. = ?**

Power

- Let β represent the chance of us making a Type II error
  - $P(\text{Fail to Reject } H_0 | H_0 \text{ False}) = P(\text{Type II Error}) = \beta$
- Therefore,
  - $P(\text{Reject } H_0 | H_0 \text{ False}) = 1 - \beta$
- **Power $= \mathbf{1} - \boldsymbol{\beta}$**
- Let H_0 corresponds to no effect
- This means H_a (or H_1) implies a significant effect
- Power is the probability of detecting a given effect if it exists

Possible outcomes of our decision

		Truth	
		H_0 True	H_0 False
Decision	Fail to Reject H_0	Correct Decision **Prob. = $1 - \alpha$**	Type II Error (False Negative) **Prob. = β**
	Reject H_0	Type I Error (False Positive) **Prob. = α**	Correct Decision **Prob. = $1 - \beta$**

Power Analysis

- We know power $= 1 - \beta$, but we cannot set both α and β
- Power is affected by
  - How big of a difference we are trying to detect
  - Variance
  - $\alpha = P(\text{Type I Error})$
  - Sample size
- Primarily focused on calculating the sample size needed for a particular power or finding the power with some sample size

Example 1

- We want to compare the effect of diet A and diet B to see which is better for weight loss
- We will do this through an experiment where half the participants are given diet A and the other half is given diet B
- $H_0: \mu_A - \mu_B = 0$ vs. $H_a: \mu_A - \mu_B \neq 0$ with $\alpha = 0.05$ and $1 - \beta \geq 0.80$
- How large does our sample size need to be to achieve a power of at least 80%?

Example 1 (cont.)

- $H_0: \mu_A - \mu_B = 0$ vs. $H_a: \mu_A - \mu_B \neq 0$ with $\alpha = 0.05$ and $1 - \beta \geq 0.80$
- Your colleague previously performed the same experiment and found $\bar{x}_A - \bar{x}_B = 1.5$ and $s_A = s_B = 1.2$
- We can use this result to determine the sample size using PROC POWER in SAS

- SAS Code:

```
PROC POWER;
        twosamplemeans test = diff
        alpha = 0.05
        stddev = 1.2
        meandiff = 1.5
        power = 0.80
        ntotal = . ;
run;
```

Example 1 (cont.)

- The code on the previous page produces the table on the right
- This tells us that given the scenario we specified, to achieve a power of at least 0.80 we need to have at least **24** people participate in the experiment

  → $N \geq 24$

The POWER Procedure
Two-Sample t Test for Mean Difference

Fixed Scenario Elements	
Distribution	Normal
Method	Exact
Alpha	0.05
Mean Difference	1.5
Standard Deviation	1.2
Nominal Power	0.8
Number of Sides	2
Null Difference	0
Group 1 Weight	1
Group 2 Weight	1

Computed N Total	
Actual Power	N Total
0.833	24

Example 1 (cont.)

- What if we have already collected the data on 20 people ($n_A = n_b = 10$) and want to know the power?
- Either set ‘ntotal = 20’ ***OR*** ‘npergroup = 10’
- Set ‘power = .’

  → $1 - \beta = 0.753$

The POWER Procedure
Two-Sample t Test for Mean Difference

Fixed Scenario Elements	
Distribution	Normal
Method	Exact
Alpha	0.05
Mean Difference	1.5
Standard Deviation	1.2
Sample Size per Group	10
Number of Sides	2
Null Difference	0

Computed Power
Power
0.753

Alternative approach to power analysis

- PROC POWER is very limited
- Does not work if a treatment design has more than one factor or has any random terms besides the residual error (e_{ij})
- In some cases, we must write our own code to determine the power

Example 2

- A donut manufacturer wants to see if the type of oil used to fry the donuts has any impact on the amount of fat absorbed by the donuts
  - 4 treatments (1, 2, 3, 4)
- The manufacturer has two types of animal fat and two types of vegetable oil that they would like to compare using four fryers to make 24 batches of donuts
  - Assume 1 and 2 are animal fat and 3 and 4 are vegetable oil
  - 6 patches per treatment

Example 2 (cont.)

- We can express the model as $y_{ij} = \mu + \tau_i + \epsilon_{ij}$
  - $i = 1,2,3,4$, $j = 1,2,\ldots,6$, and $\epsilon_{ij} \sim \mathrm{N}(0, \sigma^2)$
- The donut manufacturer found $\bar{y}_{1.} = 172$, $\bar{y}_{2.} = 185$, $\bar{y}_{3.} = 176$, $\bar{y}_{4.} = 162$, and $\sigma^2 = 100$.
  - Note that $\bar{y}_{i.} = \frac{\sum_{j=1}^{6} y_{ij}}{6}$
- How can we make our own SAS code for this?

Steps for Power Analysis

1. Generate a dataset of means
   - Set each $y_{ij} = \bar{y}_i$.
2. Run GLIMMIX with a fixed value of σ^2 to calculate the values needed to calculate power
   - Need `NumDF`, `DenDF`, and `Fvalue` from *Type III Tests of Fixed Effects* and *Contrasts* tables
   - This will allow us to find the power for each fixed effect and contrast
3. Use the results obtained in Step 2 to compute power

Example 2 – Step 1

What we need to do

- Generate a dataset of means
  - `donutpower`
- Recall
  - 4 treatments and 6 batches
  - $\bar{y}_{1.} = 172$
  - $\bar{y}_{2.} = 185$
  - $\bar{y}_{3}. = 176$
  - $\bar{y}_{4}. = 162$

SAS

```
data donutpower;
        input trt mu;
        do batch = 1 to 6;
                output;
        end;
datalines;
1 172
2 185
3 176
4 162
;
```

Example 2 – Step 1 Output

- First ten observations rows shown to the right
- 24 rows in total

Obs	trt	mu	batch
1	1	172	1
2	1	172	2
3	1	172	3
4	1	172	4
5	1	172	5
6	1	172	6
7	2	185	1
8	2	185	2
9	2	185	3
10	2	185	4

Example 2 - Step 2

What we need to do

- We need to create PROC GLIMMIX code that we would use to evaluate our model which is $y_{ij} = \mu + \tau_{ij} + \epsilon_{ij}$ with $\epsilon_{ij} \sim \mathrm{N}(0, \sigma^2)$
- Let 'y' represent y_{ij} and 'trt' represent τ_{ij}

SAS

```
proc glimmix data = <dataset>;
        class trt;
        model y = trt;
run;
```

Example 2 - Step 2 (cont.)

What we need to do

- Now we need to adapt the code from the previous slide to the dataset made in Step 1
- Replace ‘y’ with ‘mu’
- Also need the contrast the donut provider is interested in

SAS

```
proc glimmix data = donutpower;
        class trt;
        model mu = trt;
        contrast 'Animal Fat vs
              Vegetable Oil' trt 1 1 -1 -1;
run;
```

Example 2 - Step 2 (cont.)

What we need to do

- We need to extract the *Type III Tests of Fixed Effects* and *Contrasts* tables from the output
- Also need to account for $\epsilon_{ij} \sim \mathrm{N}(0,100)$

SAS

```
proc glimmix data = donutpower;
        class trt;
        model mu = trt;
        parms (100)/hold = 1;
        contrast 'Animal Oil vs
                Vegetable Oil' trt 1 1 -1 -1;
ods output contrasts=con tests3=fix;
run;
```

Example 2 – Step 2 Output (of interest)

Type III Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
trt	3	20	5.46	0.0066

Contrasts				
Label	Num DF	Den DF	F Value	Pr > F
Animal Fat vs Vegetable Oil	1	20	5.42	0.0306

Example 2 – Step 3

- Recall, power $= P(\text{Reject } H_0 | H_0 \text{ False})$
- In the even H_0 is false, the F ratio follows a non-central F distribution with the numerator degrees of freedom ('numdf') and denominator degrees of freedom ('dendf') that we obtained in Step 2, along with a non-centrality parameter, λ
- $\lambda =$'numdf' $*$ 'fvalue' → 'lambda = numdf*fvalue' in SAS
- This will be applied for each fixed effect and contrast

Example 2 – Step 3 (cont.)

- The critical region for determining the power depends on ‘numdf’, ‘dendf’ and α
  - SAS: fcrit = finv(1-alpha, numdf, dendf, 0);
- We need to set a level of alpha though.
- In SAS, we could make a DO loop that accounts for multiple levels of α
  - SAS: DO alpha = 0.10, 0.05, 0.01;
  - The code for ‘lambda’ and ‘fcrit’ go inside the DO loop

Example 2 – Step 3 (cont.)

- Finally, we can calculate $P(\text{Type II Error}) = \beta$ as a function of 'fcrit', 'numdf', 'dendf', and 'lambda' and subtract the result from 1 since power $= 1 - \beta$
  - SAS: power = 1 - probf(fcrit, numdf, dendf, lambda);
- This will also run inside the DO loop if we are interested in multiple α values

Example 2 – Step 3 (cont.)

Full SAS Code

```
data poweranalysis;
        set fix con;
        do alpha = 0.10, 0.05, 0.01;
                lambda = numdf*fvalue; * Noncentrality parameter;
                fcrit = finv(1-alpha, numdf, dendf, 0);
                power = 1 - probf(fcrit, numdf, dendf, lambda);
                output;
        end;
```

Example 2 – Step 3 Final Output

Obs	Effect	NumDF	DenDF	FValue	ProbF	Label	alpha	lambda	fcrit	power
1	trt	3	20	5.46	0.0066		0.10	16.365	2.38009	0.94158
2	trt	3	20	5.46	0.0066		0.05	16.365	3.09839	0.88195
3	trt	3	20	5.46	0.0066		0.01	16.365	4.93819	0.66653
4		1	20	5.42	0.0306	Animal Fat vs Vegetable Oil	0.10	5.415	2.97465	0.72657
5		1	20	5.42	0.0306	Animal Fat vs Vegetable Oil	0.05	5.415	4.35124	0.60044
6		1	20	5.42	0.0306	Animal Fat vs Vegetable Oil	0.01	5.415	8.09596	0.33017

Example 3

- We are interested in the effect of two different inoculation methods on Salmonella populations in beef jerky of different thicknesses. The variables involved were:
  - Salmonella counts (response variable)
  - Inoculation methods (dry and wet)
  - Thickness (¼ inch and ⅛ inch)
  - Time (weeks 1-5)
  - 5 batches for each treatment combination
- Dataset should contain 100 rows

Example 3 - Model

- 2x2 factorial design with repeated measures over time
- $Y_{ijkl} = \mu + \alpha_i + \beta_j + \tau_k + (\alpha\beta)_{ij} + (\alpha\tau)_{ik} + (\beta\tau)_{jk} + (\alpha\beta\tau)_{ijk} + u_l + e_{ijkl}$

  - Y_{ijkl}: Salmonella Level
  - μ: overall mean
  - α_i :the effect of the ith inoculation method
  - β_j: the effect of the jth thickness level
  - τ_k: r the effect of the kth week
  - $(\alpha\beta)_{ij}$: interaction effect of the ith inoculation method and the jth thickness level
  - $(\alpha\beta\tau)_{ijk}$: three-way interaction between all fixed effects
  - u_l: random effect for batches
  - e_{ijkl} : residuals

Example 3 - Contrasts

- We want to look at the contrasts between dry and wet inoculation at both 1/4 and 1/8 inches
- Will need to be accounted for in GLIMMIX statment using code below
  - contrast 'Dry vs Wet at 1/4 Inches' Inoculation_Method 1 -1 Inoculation_Method*Thickness 1 0 -1 0;
  - contrast 'Dry vs Wet at 1/8 Inches' Inoculation_Method 1 -1 Inoculation_Method*Thickness 0 1 0 -1;

Example 3 – Previous Study Results

- The table below contains the mean of the treatment combinations across the batches → $\bar{Y}_{ijk.} = \frac{\sum_{l=1}^{5} Y_{ijkl}}{5}$

Inoculation Method	Thickness	Week				
		1	2	3	4	5
Dry	1/4	4.26	4.25	4.47	4.33	4.54
	1/8	4.91	4.95	4.67	4.56	4.97
Wet	1/4	4.21	4.57	4.65	4.49	4.38
	1/8	4.86	4.78	4.62	4.32	4.22

Example 3 – Step 1

What we need to do

- Generate a dataset of means
  - `rptm_meansr`
- First we create the table from the previous slide in long format

SAS

```
data rptm_means;
        input Inoculation_Method $
Thickness $ @@;
        do Week=1 to 5 by 1;
        input mu @@;
        output;
        end;
datalines;
Dry 1/4 4.26 4.25 4.47 4.33 4.54
Dry 1/8 4.91 4.95 4.67 4.56 4.97
Wet 1/4 4.21 4.57 4.65 4.49 4.38
Wet 1/8 4.86 4.78 4.62 4.32 4.22
;
```

Example 3 – Step 1 (cont.)

What we need to do

- Now we need add in the 5 batches using a DO loop
- This will be store in ‘rptm_design’

SAS

```
data rptm_design;
        set rptm_means;
        do Batches =1 to 5;
        output;
end;
run;

proc print data=rptm_design;
run;
```

Example 3 – Step 1 Output

- First 10 rows

Obs	Inoculation_Method	Thickness	Week	mu	Batches
1	Dry	1/4	1	4.26	1
2	Dry	1/4	1	4.26	2
3	Dry	1/4	1	4.26	3
4	Dry	1/4	1	4.26	4
5	Dry	1/4	1	4.26	5
6	Dry	1/4	2	4.25	1
7	Dry	1/4	2	4.25	2
8	Dry	1/4	2	4.25	3
9	Dry	1/4	2	4.25	4
10	Dry	1/4	2	4.25	5

Example 3 – Step 2 – Fixed Effects

- Recall, the model for this example is

$$Y_{ijkl} = \mu + \alpha_i + \beta_j + \tau_k + (\alpha\beta)_{ij} + (\alpha\tau)_{ik} + (\beta\tau)_{jk} + (\alpha\beta\tau)_{ijk} + u_l + e_{ijkl}$$

- This includes all possible interactions between the fixed effects (inoculation method, thickness, and week) and all the fixed effects are categorical
  - class Batches Inoculation_Method Thickness Week;
  - model mu = Inoculation_Method|Thickness|Week;

Example 3 – Step 2 – Random Effects

$$Y_{ijkl} = \mu + \alpha_i + \beta_j + \tau_k + (\alpha\beta)_{ij} + (\alpha\tau)_{ik} + (\beta\tau)_{jk} + (\alpha\beta\tau)_{ijk} + u_l + e_{ijkl}$$

- Also includes another random term (the random effect of batches)
  - random intercept /subject=Batches;
- Additionally, the weeks are a repeated measurement
- Assume the covariance structure of the weeks is AR(1)
  - random Week/ subject=Batches*Inoculation_Method*Thickness type=ar(1) residual;

Example 3 – Step 2 – Random Effects (cont)

- We need to supply three estimates in the parms() statement and hold each for the purpose of the power analysis
  - parms(.029)(0.017)(.028)/hold=1,2,3;

Covariance Parameter Estimates			
Cov Parm	**Subject**	**Estimate**	**Standard Error**
Intercept	Batches	---	---
Variance	Batche*Inocul*Thickn	---	---
AR(1)	Batche*Inocul*Thickn	---	---

Example 3 – Step 2 Code

```
proc glimmix data=rptm_design;
        class Batches Inoculation_Method Thickness Week;
        model mu = Inoculation_Method|Thickness|Week;
        random intercept /subject=Batches;
        random Week/ subject = Batches * Inoculation_Method * Thickness type=ar(1)
residual;
        parms(.029)(0.017)(.028)/hold=1,2,3;
        contrast'Dry vs Wet at 1/4 Inches' Inoculation_Method 1 -1 Inoculation_Method *
Thickness 1 0 -1 0;
        contrast 'Dry vs Wet at 1/8 Inches' Inoculation_Method 1 -1 Inoculation_Method *
Thickness 0 1 0 -1;
ods output contrasts=f_contrast tests3=f_anova;  run;
```

Example 3 – Step 2 Output of Interest

Type III Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Inoculation_Method	1	76	9.23	0.0033
Thickness	1	76	103.28	<.0001
Inoculatio*Thickness	1	76	41.12	<.0001
Week	4	76	7.82	<.0001
Inoculation_Met*Week	4	76	13.77	<.0001
Thickness*Week	4	76	20.78	<.0001
Inocula*Thickne*Week	4	76	4.04	0.0050

Contrasts				
Label	Num DF	Den DF	F Value	Pr > F
Dry vs Wet at 1/4 Inches	1	76	5.70	0.0195
Dry vs Wet at 1/8 Inches	1	76	44.65	<.0001

Example 3 – Step 3

What we need to do

- Find the power using similar code from Example 2
- Assume $\alpha = 0.05$

SAS

```
data power;
        set f_contrast f_anova;
        alpha=0.05;
        lambda = numdf*fvalue;
        fcrit=finv(1-alpha,numdf,dendf,0);
        power=1-probf(fcrit, numdf,dendf,lambda);
run;

proc print data=power;
run;
```

Example 3 Power Output

Obs	Label	NumDF	DenDF	FValue	ProbF	Effect	alpha	lambda	fcrit	power
1	Dry vs Wet at 1/4 Inches	1	76	5.70	0.0195		0.05	5.695	3.96676	0.65406
2	Dry vs Wet at 1/8 Inches	1	76	44.65	<.0001		0.05	44.651	3.96676	1.00000
3		1	76	9.23	0.0033	Inoculation_Method	0.05	9.226	3.96676	0.85058
4		1	76	103.28	<.0001	Thickness	0.05	103.276	3.96676	1.00000
5		1	76	41.12	<.0001	Inoculatio*Thickness	0.05	41.120	3.96676	0.99999
6		4	76	7.82	<.0001	Week	0.05	31.280	2.49205	0.99659
7		4	76	13.77	<.0001	Inoculation_Met*Week	0.05	55.092	2.49205	1.00000
8		4	76	20.78	<.0001	Thickness*Week	0.05	83.132	2.49205	1.00000
9		4	76	4.04	0.0050	Inocula*Thickne*Week	0.05	16.162	2.49205	0.89525